


# Der fallende Magnet im Kupferrohr – ein Modellierungsversuch



Dr.rer.nat. Dr.techn. Martin Tutz

Jänner 2019

# 1 Einleitung

Der fallender Magnet im Kupferrohr ist ein Klassiker unter den Demonstrationsversuchen und quantitativ betrachtet trotzdem kein triviales Problem. Im Internet findet man einige qualitative Beschreibungen dieses Versuchs inkl. Animationen und Videos (siehe z.B. [2]), realitätsnahe Berechnungen liegen bislang hingegen nicht vor. Der vorliegende Artikel beschreibt einen Versuch, die Fallbewegung eines Magneten in einem Kupferrohr quantitativ nachzuvollziehen. Nach einer ausführlichen Herleitung der Bewegungsgleichung des Magneten wird die auf 2 Arten berechnete Fallgeschwindigkeit mit experimentell ermittelten Daten verglichen und diskutiert.

## 2 Herleitung


Die ersten 3 Abschnitte dienen zur Vorbereitung, auf ihre Ergebnisse wird bei der Behandlung des Magneten beim Durchfallen einer Leiterschleife zurückgegriffen. Aufbauend auf diesen Ausführungen wird abschließend der Magnet im Rohr behandelt.

### 2.1 B-Feld eines Zylindermagneten

Das Feld eines Zylindermagneten kann in Achsenrichtung ( $=\tilde{z}$ -Richtung) wie folgt beschrieben werden (siehe [1])<sup>1</sup>

$$B_M(\tilde{z}) = \frac{B_P}{2} \left( \frac{h_M + \tilde{z}}{\sqrt{r_M^2 + (h_M + \tilde{z})^2}} - \frac{\tilde{z}}{\sqrt{r_M^2 + \tilde{z}^2}} \right), \quad (1)$$

wobei  $B_P$  ein Parameter ist, der die Stärke der Magnetisierung des Magneten angibt (Herstellerangabe).  $r_M$  und  $h_M$  bezeichnen Radius und Höhe des Magneten. Die folgende Abbildung zeigt den Graphen von (1):




<sup>1</sup>Gefunden via: <https://www.supermagnete.at/faq/Wie-berechnet-man-die-magnetische-Flussdichte#formel-fr-flussdichte-zylindermagnet>

Die  $B_M$ -Funktion muss für die unten durchgeführten Betrachtungen in den Ursprung verschoben werden. Wir setzen  $\tilde{z} = z - \frac{h_M}{2}$  und erhalten nach Vereinfachung

$$B_M = B_M(z) = \frac{B_P}{2} \left( \frac{\frac{h_M}{2} + z}{\sqrt{r_M^2 + (\frac{h_M}{2} + z)^2}} - \frac{z - \frac{h_M}{2}}{\sqrt{r_M^2 + (z - \frac{h_M}{2})^2}} \right). \quad (2)$$


Stellt man (2) grafisch dar, erhält man:



Der Magnet liegt nun symmetrisch zum Koordinatenursprung.

## 2.2 Das Gesetz von Biot-Savart

Das  $\vec{B}$ -Feld eines Leiters kann mittels des Gesetzes von Biot-Savart berechnet werden. Dieses soll im Folgenden plausibilisiert werden. Wir betrachten hierzu ein von einem Strom  $I = \text{const.}$  durchflossenes Leiterstück  $d\vec{l}$  und einen beliebigen Punkt  $P$  im Raum (siehe folgende Abbildung). Der Pfeil von  $d\vec{l}$  nach  $P$  sei mit  $\vec{r}$  bezeichnet. Die Richtung des  $\vec{B}$ -Feldes in  $P$  wird durch das Vektorkreuzprodukt  $d\vec{l} \times \vec{r}$  bestimmt. Die Stärke durch den Strom  $I$  und den Abstand  $|\vec{r}|$  von  $d\vec{l}$  zu  $P$ , genauer gesagt durch die Oberfläche  $O$  der gedachten Kugel mit Radius  $|\vec{r}|$  um  $d\vec{l}$ . Sie kann durch  $O = 4r^2\pi$  mit  $r = |\vec{r}|$  berechnet werden. Die magnetische Flussdichte ist direkt proportional zur Stromstärke  $I$  und indirekt proportional zur Oberfläche  $O$  der Kugel.



Setzt man die einzelnen Teile zusammen, ergibt sich

$$d\vec{B} = \mu_0 \underbrace{\frac{I}{4r^2\pi}}_{=0} \overbrace{d\vec{l} \times \frac{\vec{r}}{|\vec{r}|}}{=d\vec{H}}, \quad (3)$$

bzw.


$$d\vec{B} = \frac{\mu_0}{4\pi} I \frac{d\vec{l} \times \vec{r}}{r^3}, \quad (4)$$

wobei  $\vec{H}$  die magnetische Feldstärke bezeichnet. Die Gesamtflussdichte  $\vec{B}$  im Punkt  $P$  erhält durch Integration über das betrachtete Leiterstück  $C$

$$\vec{B} = \frac{\mu_0}{4\pi} I \int_C \frac{d\vec{l} \times \vec{r}}{r^3}. \quad (5)$$

### 2.3 B-Feld einer kreisförmigen Leiterschleife

Im Folgenden wird das Gesetz von Biot-Savart (siehe Gleichung (5)) zur Berechnung der Flussdichte  $\vec{B}_{LS}$  einer in der  $x$ - $y$ -Ebene liegenden kreisförmigen Leiterschleife mit Radius  $r_{LS}$  in einem Punkt  $Q$  auf der Symmetrieachse (hier  $z$ -Achse) herangezogen (siehe Skizze).



Wir parametrisieren zur Berechnung des Integrals (5) das Leiterstück  $d\vec{l}$  nach  $\varphi$ . Gemäß obiger Skizze gilt

$$\vec{r}_{LS} = \begin{pmatrix} r_{LS} \cos \varphi \\ r_{LS} \sin \varphi \\ 0 \end{pmatrix}. \quad (6)$$

Da die Drehachse in  $z$ -Richtung liegt, gilt für einen infinitesimal kleinen Drehwinkel

$$d\vec{\varphi} = \begin{pmatrix} 0 \\ 0 \\ d\varphi \end{pmatrix}. \quad (7)$$

Daraus ergibt sich für

$$d\vec{l} = d\vec{\varphi} \times \vec{r}_{LS} = \begin{pmatrix} 0 \\ 0 \\ d\varphi \end{pmatrix} \times \begin{pmatrix} r_{LS} \cos \varphi \\ r_{LS} \sin \varphi \\ 0 \end{pmatrix} = \begin{pmatrix} -r_{LS} \sin \varphi \\ r_{LS} \cos \varphi \\ 0 \end{pmatrix} d\varphi. \quad (8)$$

Weiters kann man aus der Skizze  $\vec{z}_Q = \vec{r}_{LS} + \vec{r}$  ablesen. Daraus folgt

$$\vec{r} = \vec{z}_Q - \vec{r}_{LS} = \begin{pmatrix} 0 \\ 0 \\ z \end{pmatrix} - \begin{pmatrix} r_{LS} \cos \varphi \\ r_{LS} \sin \varphi \\ 0 \end{pmatrix} = \begin{pmatrix} -r_{LS} \cos \varphi \\ -r_{LS} \sin \varphi \\ z \end{pmatrix} \quad (9)$$

und

$$r = |\vec{r}| = \sqrt{r_{LS}^2 \cos^2 \varphi + r_{LS}^2 \sin^2 \varphi + z^2} = \sqrt{r_{LS}^2 + z^2}. \quad (10)$$

Setzt man die letzten Zusammenhänge in (5) ein und führt man die Integration über eine volle Umdrehung  $[0; 2\pi]$  aus, erhält man


$$\begin{aligned} \vec{B}_{LS} &= \frac{\mu_0 I}{4\pi} \int_0^{2\pi} \frac{\begin{pmatrix} -r_{LS} \sin \varphi \\ r_{LS} \cos \varphi \\ 0 \end{pmatrix} d\varphi \times \begin{pmatrix} -r_{LS} \cos \varphi \\ -r_{LS} \sin \varphi \\ z \end{pmatrix}}{(r_{LS}^2 + z^2)^{\frac{3}{2}}} \\ &= \frac{\mu_0 I}{4\pi (r_{LS}^2 + z^2)^{\frac{3}{2}}} \int_0^{2\pi} \begin{pmatrix} z r_{LS} \cos \varphi \\ z r_{LS} \sin \varphi \\ r_{LS}^2 \underbrace{(\sin^2 \varphi + \cos^2 \varphi)}_{=1} \end{pmatrix} d\varphi \\ &= \frac{\mu_0 I}{4\pi (r_{LS}^2 + z^2)^{\frac{3}{2}}} \left[ \begin{pmatrix} -z r_{LS} \sin \varphi \\ -z r_{LS} \cos \varphi \\ r_{LS}^2 \varphi \end{pmatrix} \right]_0^{2\pi} \\ &= \frac{\mu_0 I}{4\pi (r_{LS}^2 + z^2)^{\frac{3}{2}}} \begin{pmatrix} 0 \\ 0 \\ 2\pi r_{LS}^2 \end{pmatrix}. \end{aligned} \quad (11)$$

Für die in weiterer Folge relevante  $z$ -Komponente erhält man

$$B_{LS} = B_{LS,z} = \frac{\mu_0 I 2\pi r_{LS}^2}{4\pi (r_{LS}^2 + z^2)^{\frac{3}{2}}} = \frac{\mu_0 I}{2} \frac{r_{LS}^2}{(r_{LS}^2 + z^2)^{\frac{3}{2}}}. \quad (12)$$

## 2.4 Die Bewegungsgleichung für einen Ring

Lässt man den Magneten durch einen Ring mit hinreichend kleiner Ausdehnung  $h_{LS}$  in  $z$ -Richtung fallen (siehe Skizze),



so ergibt sich für die Bewegungsgleichung der Magneten unter Vernachlässigung der Reibung auf Basis des 2. Newtonschen Gesetzes

$$ma_z = -mg + F_M, \quad (13)$$

wobei  $m$  die Masse des Magneten,  $a_z = \ddot{z}(t) = \ddot{z}$  die Beschleunigung des Magneten in  $z$ -Richtung,  $g$  die Erdbeschleunigung und  $F_M$  die magnetische Kraft bezeichnen. Letztere wird ausgehend von der bekannten Formel für die Lorentz-Kraft (skalar angeschrieben)

$$F_M = F_L = IlB \quad (14)$$

wie folgt modelliert: Das Produkt  $Il$  wird durch die magnetische Flussdichte der Leiterschleife  $B_{LS}$  am Ort des Magneten ersetzt. In  $B_{LS}$  ist der Abstand gemäß dem Gesetz von Biot-Savart bereits berücksichtigt (siehe Gleichung (5)). Allerdings muss die im letztgenannten Gesetz enthaltene Multiplikation mit der magnetischen Feldkonstante  $\mu_0$  „rückgängig“ gemacht werden. D.h., dass die Substitution eine zusätzliche Division durch  $\mu_0$  nach sich zieht. Weiters liefert das Gesetz von Biot-Savart lediglich die magnetische Flussdichte an einem einzigen Punkt. Da der Magnet jedoch eine Ausdehnung besitzt (=Querschnittsfläche  $A_M$ ), muss zur Berechnung der (Gesamt-)Kraft mit der Fläche  $A_M$  multipliziert werden. Die magnetische Flussdichte  $B$  in (14) muss jene des Magneten  $B_M$  sein. Zusammengesetzt erhält man

$$F_M = \frac{1}{\mu_0} A_M B_M B_{LS}. \quad (15)$$

**Bemerkung:** Da der Gesamtfluss des Magneten mittels  $\Phi_M = A_M B_M$  und die Feldstärke der Leiterschleife durch  $H_{LS} = \frac{1}{\mu_0} B_{LS}$  ausgedrückt werden kann, folgt für (15)

$$F_M = \Phi_M H_{LS}. \quad (16)$$

Dieser Zusammenhang findet sich auch in der offiziellen Formelsammlung der österreichischen Physikolympiade wieder.

Darüber hinaus kann die Richtigkeit der Überlegungen durch eine Dimensionsbetrachtung untermauert werden:

$$[F_M] = \frac{1}{[\mu_0]} [A_M] [B_M] [B_{LS}] \quad (17)$$

$$N = \frac{1}{\frac{V_s}{Am}} m^2 \frac{V_s V_s}{m^2 m^2} \quad (18)$$

$$\frac{J}{m} = \frac{Am}{V_s} m^2 \frac{V_s V_s}{m^2 m^2} \quad (\text{Kürzen!}) \quad (19)$$

$$\frac{VAs}{m} = \frac{VAs}{m} \quad \text{w.z.z.w.} \quad (20)$$

Im Folgenden wird (13) ausformuliert: Für  $B_M$  wird unter Zuhilfenahme von (2)

$$B_{M,0} = B_M(0) = \text{const.} \quad (21)$$

eingesetzt und für  $B_{LS}$  die Beziehung (12) mit ( $z_{LS}$ ... Höhe der Leiterschleife, siehe Skizze)

$$B_{LS} = \frac{\mu_0 I}{2} \frac{r_{LS}^2}{(r_{LS}^2 + (z - z_{LS})^2)^{\frac{3}{2}}} \quad (22)$$

substituiert. Die darin vorkommende bislang unbekannte Stromstärke  $I$  kann für eine in der Höhe  $z_{LS}$  befindliche Leiterschleife unter Zuhilfenahme des Induktionsgesetzes wie folgt berechnet werden (zur Erinnerung  $z = z(t)$ )

$$I = \frac{U}{R_{LS}} \quad (23a)$$

$$= -\frac{1}{R_{LS}} \frac{d\Phi}{dt} \quad (23b)$$

$$= -\frac{1}{R_{LS}} \frac{d(A_{LS} B_M(z - z_{LS}))}{dt} \quad (23c)$$

$$= -\frac{1}{R_{LS}} A_{LS} \frac{B_M(z - z_{LS})}{dt} \quad (23d)$$

$$= -\frac{1}{R_{LS}} A_{LS} \frac{d}{dt} \left( \frac{B_P}{2} \left( \frac{\frac{h_M}{2} + z - z_{LS}}{\sqrt{r_M^2 + (\frac{h_M}{2} + z - z_{LS})^2}} - \frac{z - z_{LS} - \frac{h_M}{2}}{\sqrt{r_M^2 + (z - z_{LS} - \frac{h_M}{2})^2}} \right) \right), \quad (23e)$$

wobei für

$$R_{LS} = \rho \frac{l_{LS}}{A_{LSQ}} \quad (24)$$

gilt.

In den obigen Ausführungen bezeichnen  $U$  die vom Magneten in der Leiterschleife induzierte Spannung,  $R_{LS}$  den Widerstand der Leiterschleife,  $\Phi$  den vom Magneten in der Leiterschleife

verursachten magnetischen Fluss,  $A_{LS}$  die vom Fluss durchsetzte Fläche der Leiterschleife,  $B_M(z - z_{LS})$  die vom Magneten verursachte Flussdichte am Ort der Leiterschleife, sowie  $\rho$  den spezifischen Widerstand der Leiterschleife,  $l_{LS} = 2\pi r_{LS}$  den Umfang der Leiterschleife (=Umfang/Länge des Ringes) und  $A_{LSQ}$  die Querschnittsfläche der Leiterschleife (siehe obige Skizze).

Setzt man alle obigen Beziehungen in (13) ein, erhält man

$$m\ddot{z} = -mg + \frac{1}{\mu_0} A_M B_{M,0} \frac{\mu_0}{2} \frac{r_{LS}^2}{(r_{LS}^2 + (z - z_{LS})^2)^{\frac{3}{2}}} \quad (25a)$$

$$\cdot \left( -\frac{1}{\rho \frac{l_{LS}}{A_{LSQ}}} A_{LS} \operatorname{sgn}(z_{LS} - z) \right) \quad (25b)$$

$$\cdot \frac{d}{dt} \left( \frac{B_P}{2} \left( \frac{\frac{h_M}{2} + z - z_{LS}}{\sqrt{r_M^2 + (\frac{h_M}{2} + z - z_{LS})^2}} - \frac{z - z_{LS} - \frac{h_M}{2}}{\sqrt{r_M^2 + (z - z_{LS} - \frac{h_M}{2})^2}} \right) \right), \quad (25c)$$

bzw. vereinfacht

$$m\ddot{z} = -mg + A_M B_{M,0} \frac{1}{2} \frac{r_{LS}^2}{(r_{LS}^2 + (z - z_{LS})^2)^{\frac{3}{2}}} \frac{1}{\rho \frac{l_{LS}}{A_{LSQ}}} A_{LS} \operatorname{sgn}(z - z_{LS}) \quad (26a)$$


$$\cdot \frac{B_P}{2} \frac{d}{dt} \left( \frac{\frac{h_M}{2} + z - z_{LS}}{\sqrt{r_M^2 + (\frac{h_M}{2} + z - z_{LS})^2}} - \frac{z - z_{LS} - \frac{h_M}{2}}{\sqrt{r_M^2 + (z - z_{LS} - \frac{h_M}{2})^2}} \right). \quad (26b)$$

Die Signumfunktion  $\operatorname{sgn}(z - z_{LS})$  gibt jene Tatsache wieder, dass sich die Stromrichtung in der Leiterschleife umdreht, wenn der Magnet diese passiert. D.h. je nachdem, ob sich der Magnet oberhalb oder unterhalb der Leiterschleife befindet, dreht der durch die induzierte Spannung verursachte Strom in unterschiedliche Richtungen.

Die bislang angestellten Überlegungen werden im folgenden Abschnitt für die Beschreibung des Magneten in einem Rohr erweitert.

## 2.5 Die Bewegungsgleichung für ein Rohr

Für die Berechnung der Fallbewegung eines Magneten in einem ausgedehnten Rohr mit der Länge  $l_{ges}$ , stellt man sich das Rohr in einzelne Leiterschleifen/Ringe aufgeteilt vor (siehe Skizze).



Bei einer (hinreichend kleinen) Leiterschleifenhöhe  $h_{LS}$  ergeben sich somit

$$N_{LS} = \frac{l_{ges}}{h_{LS}} \quad (27)$$

Ringe. Für jede einzelne (gedachte) Leiterschleife gilt (26). Wir ersetzen  $z_{LS}$  durch  $z_{LS,i}$  (=Höhe der Ringmitten, siehe Skizze) und summieren zur Berechnung der Gesamtkraft über alle Leiterschleifen  $i = 1, \dots, N_{LS}$ . Vereinfacht erhält man

$$m\ddot{z} = -mg + A_M B_{M,0} A_{LS} \frac{1}{2} \frac{1}{\rho \frac{l_{LS}}{A_{LSQ}}} \sum_{i=1}^{N_{LS}} \frac{r_{LS}^2}{(r_{LS}^2 + (z - z_{LS,i})^2)^{\frac{3}{2}}} \operatorname{sgn}(z - z_{LS,i}) \quad (28a)$$

$$\cdot \frac{B_P}{2} \frac{d}{dt} \left( \frac{\frac{h_M}{2} + z - z_{LS,i}}{\sqrt{r_M^2 + (\frac{h_M}{2} + z - z_{LS,i})^2}} - \frac{z - z_{LS,i} - \frac{h_M}{2}}{\sqrt{r_M^2 + (z - z_{LS,i} - \frac{h_M}{2})^2}} \right). \quad (28b)$$

## 3 Lösungen

### 3.1 Numerische Lösung der Bewegungsgleichung

Bei (28) handelt es sich um eine gewöhnliche nichtlineare Differentialgleichung (DGL) 2. Ordnung mit  $z = z(t)$ , die numerisch gelöst werden kann. Dividiert man (28) durch  $m$ ,

erhält man eine der DGL der Bauform

$$\ddot{z} = -g + \frac{1}{m} A_M B_{M,0} A_{LS} \frac{1}{2} \frac{1}{\rho_{A_{LSQ}}^{l_{LS}}} \sum_{i=1}^{N_{LS}} \frac{r_{LS}^2}{(r_{LS}^2 + (z - z_{LS,i})^2)^{\frac{3}{2}}} \text{sgn}(z - z_{LS,i}) \quad (29a)$$

$$\cdot \frac{B_P}{2} \frac{d}{dt} \left( \frac{\frac{h_M}{2} + z - z_{LS,i}}{\sqrt{r_M^2 + (\frac{h_M}{2} + z - z_{LS,i})^2}} - \frac{z - z_{LS,i} - \frac{h_M}{2}}{\sqrt{r_M^2 + (z - z_{LS,i} - \frac{h_M}{2})^2}} \right). \quad (29b)$$

Wir wählen dem Experiment entsprechend die zwei Startwerte

$$z(t=0) = 0 \text{ und} \quad (30)$$

$$\dot{z}(t=0) = \dot{z}_0 = 0. \quad (31)$$

Auf einem äquidistanten Gitter mit der Schrittweite  $\Delta t$ , d.h.  $t_j = j\Delta t$ ,  $j = 0, 1, 2, \dots, N_t$  kann der Startschritt mittels

$$\ddot{z}(t_0) \approx \frac{z_1 - z_0 - \dot{z}_0}{\frac{\Delta t}{2}} \quad (32)$$

diskretisiert werden. Eingesetzt in (29) erhält man vereinfacht und umgeformt (Beachte, dass aufgrund von  $\dot{z}_0 = 0$  für  $F_M = 0$  folgt!)

$$z_1 = z_0 - \frac{\Delta t^2}{2} g. \quad (33)$$

Für  $j \geq 2$  gilt

$$\ddot{z}(t_{j-1}) \approx \frac{z_j - 2z_{j-1} + z_{j-2}}{\Delta t^2}. \quad (34)$$

Von (29) gelangt man vollständig diskretisiert<sup>2</sup> und umgeformt zu

$$z_j = 2z_{j-1} - z_{j-2} + \Delta t^2 \left( -g \right. \quad (35a)$$

$$\left. + \frac{1}{m} A_M B_{M,0} A_{LS} \frac{1}{2} \frac{1}{\rho_{A_{LSQ}}^{l_{LS}}} \sum_{i=1}^{N_{LS}} \frac{r_{LS}^2}{(r_{LS}^2 + (z_{j-1} - z_{LS,i})^2)^{\frac{3}{2}}} \text{sgn}(z_{j-1} - z_{LS,i}) \right. \quad (35b)$$

$$\left. \cdot \frac{B_P}{2} \frac{1}{\Delta t} \left( \left( \frac{\frac{h_M}{2} + z_{j-1} - z_{LS,i}}{\sqrt{r_M^2 + (\frac{h_M}{2} + z_{j-1} - z_{LS,i})^2}} - \frac{z_{j-1} - z_{LS,i} - \frac{h_M}{2}}{\sqrt{r_M^2 + (z_{j-1} - z_{LS,i} - \frac{h_M}{2})^2}} \right) \right. \quad (35c)$$

$$\left. - \left( \frac{\frac{h_M}{2} + z_{j-2} - z_{LS,i}}{\sqrt{r_M^2 + (\frac{h_M}{2} + z_{j-2} - z_{LS,i})^2}} - \frac{z_{j-2} - z_{LS,i} - \frac{h_M}{2}}{\sqrt{r_M^2 + (z_{j-2} - z_{LS,i} - \frac{h_M}{2})^2}} \right) \right) \right). \quad (35d)$$

Ein Beispiel für eine konkrete Simulation zeigt Abschnitt 4.

<sup>2</sup>Da keine starke Dämpfung vorhanden ist, wurde ein explizites Verfahren gewählt.

### 3.2 Analytische Berechnung der konstanten Fallgeschwindigkeit im Rohr

Ist der Magnet hinreichend weit von den Enden des Rohres entfernt, kann die weitgehend konstante(!) Fallgeschwindigkeit  $v = \dot{z}$  des Magneten auch analytisch berechnet werden. Für  $v = const.$  vereinfacht sich (28) nach dem 1. Newtonschen Gesetz zu

$$mg = A_M B_{M,0} A_{LS} \frac{1}{2} \frac{1}{\rho \frac{l_{LS}}{A_{LSQ}}} \sum_{i=1}^{N_{LS}} \frac{r_{LS}^2}{(r_{LS}^2 + (z - z_{LS,i})^2)^{\frac{3}{2}}} \operatorname{sgn}(z - z_{LS,i}) \quad (36a)$$

$$\cdot \frac{B_P}{2} \frac{d}{dt} \left( \frac{\frac{h_M}{2} + z - z_{LS,i}}{\sqrt{r_M^2 + (\frac{h_M}{2} + z - z_{LS,i})^2}} - \frac{z - z_{LS,i} - \frac{h_M}{2}}{\sqrt{r_M^2 + (z - z_{LS,i} - \frac{h_M}{2})^2}} \right). \quad (36b)$$

Mit  $z = z(t)$  kann die Ableitung der Flussdichtefunktion nach der Zeit  $t$  durch fortgesetztes Differenzieren ausgeführt werden. Setzt man anschließend für  $z$  die Mitte des Rohres  $z = \frac{l_{ges}}{2}$  ein, kann man  $v = \dot{z}$  als einzige verbleibende Variable aus der Summe (bestehend aus zur Gänze bekannten Termen) herausheben und man erhält eine Gleichung der Bauform


$$mg = v \sum_{i=1}^{N_{LS}} (\dots), \quad (37)$$

die nach  $v$  aufgelöst werden kann<sup>3</sup>.

Ein Beispiel für eine konkrete Berechnung der Geschwindigkeit  $v$  zeigt Abschnitt 4.


## 4 Vergleich von Berechnung und Messung – Diskussion

Für die Überprüfung der Modellbildung wurde eine Simulation durchgeführt, deren Parameter dem Programm aus Anhang A entnommen werden können. Die folgenden Diagramme zeigen die Weg-Funktion ( $z$ -Koordinate)



<sup>3</sup>Zur Durchführung einer konkreten Berechnung ist die Verwendung eines CAS empfohlen.

sowie die Geschwindigkeitsfunktion aufgetragen über der Zeit  $t$ .



Das Geschwindigkeitsdiagramm zeigt in der Rohrmitte eine Geschwindigkeit von  $v_{ger,num} \approx -0,42 \frac{m}{s}$ . Vergleicht man diese mit dem gemessenen Wert (siehe unterer Teil des Quellcodes aus Anhang A)  $v_{gem} \approx -0,45 \frac{m}{s}$ , so ergibt sich ein relativer Fehler bezogen auf den Messwert von  $\Delta v_{rel,num,\%} \approx -5,4\%$ .

Ein Vergleich mit der analytisch berechneten Geschwindigkeit (siehe Anhang B) liefert:  $v_{ger,ana} \approx -0,43 \frac{m}{s}$  und somit einen relativen Fehler von  $\Delta v_{rel,ana,\%} \approx -4,7\%$ .

Beide Berechnungsvarianten zeigen somit eine akzeptable Übereinstimmung mit den Experimentaldaten. Der Verlauf von  $z$  und  $\dot{z}$  erscheint plausibel, insbesondere werden auch die Randbereiche korrekt wiedergegeben. Der Unterschied bei den relativen Fehlern (analytisch und numerisch) lässt sich darauf zurückführen, dass bei der analytischen Berechnung Randeffekte am oberen und am unteren Ende des Rohres nicht berücksichtigt werden.

Anzumerken ist hierbei allerdings, dass diese gute Übereinstimmung nur erreicht wird, wenn die Querschnittsfläche des Magneten  $A_M$  die Querschnittsfläche des Rohres  $A_{LS}$  „gut ausfüllt“. Ist der Radius des Magneten wesentlich kleiner als jener des Rohres, so sind die relativen Fehler wesentlich größer. Dies liegt daran, dass der die Spannung induzierende magnetische Fluss des Magneten in der jeweiligen Leiterschleife bei der gewählten Modellierung überschätzt wird. Darüber hinaus wird die magnetische Flussdichte einer Leiterschleife nur für die Mitte des Rohres im Abstand des Magneten berechnet (Biot-Savart), der Fluss jedoch aufgrund der Multiplikation mit der Fläche des Magneten  $A_M$  für den kompletten Querschnitt zu hoch modelliert (fehlende Randeffekte).

# A Anhang – Quellcode des Maxima-Programms 1

Es folgt der Quellcode des Maxima-Programms zur numerischen Lösung der Bewegungsgleichung aus Abschnitt 3.1.

```
(%i1) kill(all);
(%o0) done
Magnet im Kupferrohr
Alle Einheiten in SI!
Autor: Martin Tutz
(%i1) ratprint:false$
(%i2) load(mnewton)$
Masse des (Zylinder-)Magneteten
(%i3) m:0.0086;
(m) 0.0086
Erdbeschleunigung
(%i4) g:9.81;
(g) 9.81
Radius des Magneteten
(%i5) rm:0.006;
(rm) 0.006
Querschnittsfläche des Magneteten
(%i6) Am:float(rm^2*pi);
(Am) 1.1309733552923255*10^-4
Höhe des Magneteten
(%i7) hm:0.01;
(hm) 0.01
Koeff. der magnetischen Flussdichte
(%i8) Bm:1.35;
(Bm) 1.35
'Ungefähre' Reichweite des Magneteten (siehe B-Feld-Funktion)
(%i9) rwm:0.05;
(rwm) 0.05
Radius des Rohres
(%i10) rr:0.007;
(rr) 0.007
Länge des Rohres
(%i11) lr:0.05;
(lr) 0.05
Querschnittsfläche des Rohres
(%i12) Ar:float(rr^2*pi);
(Ar) 1.539380400258999*10^-4
Umfangs des Rohres
(%i13) Ur:float(rr*pi);
(Ur) 0.021991148575128
Wandstärke des Kupferrohres
(%i14) di:0.001;
(di) 0.001
```

```

Zeitschrittweite für die numerische Integration
(%i15) deltat:0.001;
(deltat) 0.001
Höhe eines Ringes des Rohres (Diskretisierung des Rohres)
(%i16) deltaz:0.001;
(deltaz) 0.001
Anzahl der Rohrabschnitte/Ringe
(%i17) Nr:round(lr/deltaz);
(Nr) 50
Querschnittsfläche eines Ringes
(%i18) Aqr:deltaz*di;
(Aqr) 1.0*10^-6
Spez. Widerstand von Kupfer
(%i19) rhocu:1.8*10^(-8);
(rhocu) 1.8000000000000002*10^-8
Elektrischer Widerstand eines Ringes
(%i20) Rring:rhocu*Ur/Aqr;
(Rring) 3.95840674352314*10^-4
Berechnung der absoluten Höhen der einzelnen Ringe
(%i21) for ii:1 thru Nr do block(
    hr[ii]:deltaz/2.+(ii-1)*deltaz
);
(%o21) done
Magnetische Feldkonstante
(%i22) mu0:float(4*pi*10^(-7));
(mu0) 1.2566370614359173*10^-6
Flussdichtefunktion des Zylindermagneten
(%i23) Bges(x):=' '(Bm/2*((hm+x-hm/2)/(sqrt(rm^2
+(hm+x-hm/2)^2))-(x-hm/2)/sqrt(rm^2+(x-hm/2)^2)));
/*Bges(x):=' '(Bm/2*((hm-(x-hm/2))/(sqrt(rm^2
+(hm-(x-hm/2)^2))-(x-hm/2)/sqrt(rm^2+(x-hm/2)^2)));*/
/*Bges(x):=' '(Bm/2*((hm-x)/(sqrt(rm^2
+(hm-x)^2))-x/sqrt(rm^2+x^2)));*/
(%o23) Bges(x):=0.675*((x+0.005)/sqrt((x+0.005)^2+3.6*10^-5))
-(x-0.005)/sqrt((x-0.005)^2+3.6*10^-5))
(%i24) Bmagn:Bges(0);
(Bmagn) 0.86424893954704
Reibungskoeffizient
(%i25) mu_Reibung:0.15;
(mu_Reibung) 0.15
Normalkraft
(%i26) FN:m*g/10;
(FN) 0.0084366
Plot der Flussdichtefunktion
(%i27) wxplot2d([Bges], [x,-rwm,rwm])$
(%t27) (Bild)
Startwert Ort

```

```

(%i28) z[0]:lr;
(%o28) 0.05
Startwert Geschwindigkeit
(%i29) v[0]:0.;
(%o29) 0
Startschritt
(%i30) z[1]:z[0]+v[0]*deltat-g*deltat^2/2;
(%o30) 0.049995095
(%i31) i:2;
(i) 2
Iteration
(%i32) while z[i-1]>0. do block(
  z[i]:2*z[i-1]-z[i-2]+deltat^2*1/m*(-m*g
+float(1)/mu0*Am*Bmagn*mu0/2*1/Rring*Ar
*sum(signum(z[i-1]-hr[ih])*(Bges(z[i-1]-hr[ih])
-Bges(z[i-2]-hr[ih])))/deltat
*rr^2/(rr^2+(z[i-1]-hr[ih])^2)^(1.5),ih,1,Nr)),
  if mod(i,1000)=0 then display(i,z[i]),
  i:i+1
);
i=1000
z[1000]=0.0048583166674324
(%o32) done
Plot der Ortsfunktion (z-Koordinate)
(%i33) zliste:makelist([iii*deltat,z[iii]],iii,0,i-1)$
(%i34) wxplot2d([discrete,zliste])$
(%t34) (Bild)
(%i35) plot2d([discrete,zliste],[ylabel,"Weg z in m"],
[xlabel,"Zeit t in s"],[gnuplot_term,ps],[gnuplot_out_file,
"/home/martin/Mathe-Physik/Fallrohr-aus-Kupfer/z.eps"]);
(%o35) "/home/martin/maxout.gnuplot_pipes
/home/martin/Mathe-Physik/Fallrohr-aus-Kupfer/z.eps"
Plot der Geschwindigkeitsfunktion
(%i36) for iv:0 thru i-2 do v[iv]:(z[iv+1]-z[iv])/deltat;
(%o36) done
(%i37) vliste:makelist([iii*deltat,v[iii]],iii,0,i-2)$
(%i38) v_Mitte:v[round((i-2)/2)];
(v_Mitte) -0.042283077629656
(%i39) wxplot2d([discrete,vliste])$
(%t39) (Bild)
(%i40) plot2d([discrete,vliste],[ylabel,"Geschwindigkeit v in m/s"],
[xlabel,"Zeit t in s"],[gnuplot_term,ps],[gnuplot_out_file,
"/home/martin/Mathe-Physik/Fallrohr-aus-Kupfer/v.eps"]);
(%o40) "/home/martin/maxout.gnuplot_pipes
/home/martin/Mathe-Physik/Fallrohr-aus-Kupfer/v.eps"
(%i41) t_gemessen:(23.37+23.94)/2;
(t_gemessen) 23.655

```

```
(%i42) laenge:-1.058;
(laenge) -1.058
(%i43) v_exp:laenge/t_gemessen;
(v_exp) -0.044726273515113
(%i44) Deltat_rel:(v_Mitte-v_exp)/v_exp;
(Deltat_rel) -0.054625518592125
```

## B Anhang – Quellcode des Maxima-Programms 2

Es folgt der Quellcode des Maxima-Programms zur Berechnung der konstanten Fallgeschwindigkeit des Magneten aus Abschnitt 3.2.

```
(%i1) kill(all);
(%o0) done
Magnet im Kupferrohr
Alle Einheiten in SI!
Autor: Martin Tutz
(%i1) ratprint:false$
(%i2) load(mnewton)$
Masse des (Zylinder-)Magneten
(%i3) m:0.0086;
(m) 0.0086
Erdbeschleunigung
(%i4) g:9.81;
(g) 9.81
Radius des Magneten
(%i5) rm:0.006;
(rm) 0.006
Querschnittsfläche des Magneten
(%i6) Am:float(rm^2*%pi);
(Am) 1.1309733552923255*10^-4
Höhe des Magneten
(%i7) hm:0.01;
(hm) 0.01
Koeff. der magnetischen Flussdichte
(%i8) Bm:1.35;
(Bm) 1.35
'Ungefähre' Reichweite des Magneten (siehe B-Feld-Funktion)
(%i9) rwm:0.05;
(rwm) 0.05
Radius des Rohres
(%i10) rr:0.007;
(rr) 0.007
Länge des Rohres
(%i11) lr:1.057;
(lr) 1.057
Querschnittsfläche des Rohres
```

```

(%i12) Ar:float(rr^2*pi);
(Ar) 1.539380400258999*10^-4
  Umfangs des Rohres
(%i13) Ur:float(rr*pi);
(Ur) 0.021991148575128
  Wandstärke des Kupferrohres
(%i14) di:0.001;
(di) 0.001
  Höhe eines Ringes des Rohres (Diskretisierung des Rohres)
(%i15) deltaz:0.001;
(deltaz) 0.001
  Anzahl der Rohrabschnitte/Ringe
(%i16) Nr:round(lr/deltaz);
(Nr) 1057
(%i17) Mitte:round(Nr/2);
(Mitte) 528
  Querschnittsfläche eines Ringes
(%i18) Aqr:deltaz*di;
(Aqr) 1.0*10^-6
  Spez. Widerstand von Kupfer
(%i19) rhocu:1.8*10^(-8);
(rhocu) 1.8000000000000002*10^-8
  Elektrischer Widerstand eines Ringes
(%i20) Rring:rhocu*Ur/Aqr;
(Rring) 3.95840674352314*10^-4
  Berechnung der absoluten Höhen der einzelnen Ringe
(%i21) for ii:1 thru Nr do block(
  hr[ii]:deltaz/2.+(ii-1)*deltaz
);
(%o21) done
  Magnetische Feldkonstante
(%i22) mu0:float(4*pi*10^(-7));
(mu0) 1.2566370614359173*10^-6
  Flussdichtefunktion des Zylindermagneten
(%i23) Bges(x):=' '(Bm/2*((hm+x-hm/2)/(sqrt(rm^2+(hm+x-hm/2)^2))
-(x-hm/2)/sqrt(rm^2+(x-hm/2)^2)));
(%o23) Bges(x):=0.675*((x+0.005)/sqrt((x+0.005)^2+3.6*10^-5)
-(x-0.005)/sqrt((x-0.005)^2+3.6*10^-5))
(%i24) Bges(z(t)-hrx);
(%o24) 0.675*((z(t)-hrx+0.005)/sqrt((z(t)-hrx+0.005)^2+3.6*10^-5)
-(z(t)-hrx-0.005)/sqrt((z(t)-hrx-0.005)^2+3.6*10^-5))
(%i25) Bdt:diff(%,t);
(Bdt) 0.675*(diff(z(t),t,1)/sqrt((z(t)-hrx+0.005)^2+3.6*10^-5)
-((z(t)-hrx+0.005)^2*(diff(z(t),t,1)))/((z(t)-hrx+0.005)^2
+3.6*10^-5)^(3/2))-diff(z(t),t,1)/sqrt((z(t)-hrx-0.005)^2+3.6*10^-5)
+((z(t)-hrx-0.005)^2*(diff(z(t),t,1)))/((z(t)-hrx-0.005)^2+3.6*10^-5)^(3/2))
(%i26) subst(v,diff(z(t),t),%);

```

```

(%o26) 0.675*(v/sqrt((z(t)-hrx+0.005)^2+3.6*10^-5)
-((z(t)-hrx+0.005)^2*v)/((z(t)-hrx+0.005)^2+3.6*10^-5)^(3/2)
-v/sqrt((z(t)-hrx-0.005)^2+3.6*10^-5)+((z(t)-hrx-0.005)^2*v)/
((z(t)-hrx-0.005)^2+3.6*10^-5)^(3/2))
(%i27) Bgesdt(hrx):=' '(subst(hr[Mitte],z(t),%));
(%o27) Bgesdt(hrx):=0.675*(-((0.5325-hrx)^2*v)/((0.5325-hrx)^2
+3.6*10^-5)^(3/2)+((0.5225-hrx)^2*v)/((0.5225-hrx)^2+3.6*10^-5)^(3/2)
+v/sqrt((0.5325-hrx)^2+3.6*10^-5)-v/sqrt((0.5225-hrx)^2+3.6*10^-5))
Plot der Flussdichtefunktion
(%i28) wxplot2d([Bges], [x,-rwm,rwm])$
(%t28) (Bild)
(%i29) Bmagn:Bges(0);
(Bmagn) 0.86424893954704
(%i30) Bpol:Bges(hm/2);
(Bpol) 0.57880772485596
(%i31) glg:m*g=float(1)/mu0*Am*Bmagn*mu0/2
*1/Rring*Ar*sum(signum(hr[Mitte]-hr[ih])*Bgesdt(hr[ih])
*rr^2/(rr^2+(hr[Mitte]-hr[ih])^2)^(1.5),ih,1,Nr);
(glg) 0.084366=-1.981857576735746*v
(%i32) loes[i]:mnewton(glg,v,0);
(%o32) [[v=-0.042569153803148]]
(%i33) t_gemessen:(23.37+23.94)/2;
(t_gemessen) 23.655
(%i34) v_exp:-lr/t_gemessen;
(v_exp) -0.044683999154512
(%i35) v_rech:rhs(loes[i][1][1]);
(v_rech) -0.042569153803148
(%i36) Deltat_rel:(v_rech-v_exp)/v_exp;
(Deltat_rel) -0.047328918435678

```

## Literatur

- [1] J.M. Camacho, V. Sosa, *Alternative method to calculate the magnetic field of permanent magnets with azimuthal symmetry*, Revista de mecanica de fisica, E59, S. 8-17, 2013, online: <https://www.researchgate.net/publication/262707508>, 28.12.2018.
- [2] J. Maxton-Küchenmeister et. al., *Fallende Magnete*, online: <https://www.leifiphysik.de/elektrizitaetslehre/elektromagnetische-induktion/versuche/fallende-magnete>, 20.1.2019.